
CASOS ECOMET  
  
Instrucciones manejo ECOMET  
  Abrir ficheros 
  Para cada caso 

a) Abrir el caso en Abrir archivo de sentencias  del menú Util.  
 b) Abrir el mismo caso, si no se especifica otra cosa en el archivo de sentencias, en Abrir archivo 
de trabajo  del menú Archivo. 
 
Ejecutar sentencias 

a) Seleccionar con el ratón las sentencias que se desean ejecutar 
b) Activar la opción Procesar en el menú de Proced. 
Se debe procurar seguir un orden secuencial, ya que la ejecución de algunas sentencias puede 
implicar que se hayan ejecutado algunas sentencias previamente 

 
Visualizar matrices 

a) Seleccionar la opción Editor de matrices en el menu  Datos 
b) Teclear el código de la matriz 
c) Pulsar la tecla Intro 
d) Activar la opción Mostrar matriz 
Para salir del Editor de matrices seleccionar la opción Salir en Miscelánea 

 
 
; CASO 001 

 
 ; generacion de una serie  
 ; poblacion normal media 170 y desviacion tipica 10  
muest 1 500 
ser enum = 170 + rndn*10 
hist enum miny 0  maxy 200   minx 140 maxx 200  interx  10   intery  1 
esc mediaser= medser(enum)  
esc destiser=desser(enum) 
visesc mediaser 
visesc destiser 
 
 ; generacion de 100 series  
muest 1 100 
ser mediam=0 
deci 0 
muest 1  20 
repite  j desde 1 hasta 100 
esc mediaser=0 
ser enum = 170 + rndn*10 
esc mediaser= medser(enum) 
obs mediam[j]=mediaser 
finrepite 
 
 ; Calculo de la MEDIA y DESVIACION TIPICA  global 
 ; media teorica 170  desviacion tipica 10/raizc(20)=2.24 
muest 1 100 
hist mediam    minx 160 maxx 180 interx  10   intery  1 
esc mediaglo =medser(mediam) 
esc desviglo =desser(mediam) 
visesc mediaglo 
visesc desviglo  
 
  
 ; CASO 208 
 



ser ipci=ipc*100/67.4 
ser salmini=salmin*100/750 
ser salminv=(salmin-salmin(-1))*100/salmin(-1) 
ser salmind=salmin*100/ipc 
ser ipcv=(ipc-ipc(-1))*100/ipc(-1) 
 
 ; graficos de series 
graf salmin ipc 
graf salmini ipci 
graf salminv ipcv 
 
 ; grafico de dispersion y estimacion MC 
graf (ipc) salmin 
mc salmin 1 ipc 
 
 ; calculo de la regresion utilizando algebra ordinaria 
ser Yd=salmin-medser(salmin) 
ser Xd=ipc-medser(ipc) 
ser YdXd=Yd*Xd 
ser Xd2=Xd*Xd 
esc beta=sumser(YdXd)/sumser(Xd2) 
esc alfa=medser(salmin)-beta*medser(ipc) 
visesc beta 
visesc alfa 
 
 
 ; CASO 301/304 
 
 ; CASO 301 
 
muest 1 20 
deci 0 
 
 ; grafico dispersion y estimacion MC 
graf (valcon)  valacc 
mc valacc 1 valcon 
 
              ; calculo de coeficientes  
mat x ={ 1 valcon} 
mat y= {valacc} 
 
mat xx=x'x 
mat xy=x'y 
mat xxinv=xx$ 
mat coef =xxinv*xy 
mat coef1=(x'x)$x'y 
 
 ; calculo del vector de residuos 
mat u=y-x*coef1 
 
 ; calculo del estimador de la varianza de las perturbaciones 
mat uu=u'u 
esc uu2=escal(uu) 
mat uu1=y'y-coef'x'y 
esc uu22=escal(uu1) 
esc varu=uu2/(nobs-_nreg) 
esc desu=rcuad(varu) 
 
               ; calculo de la matriz var-cov de los estimadores 
mat covacoef=varu*(x'x)$ 



 
 ; calculo de la media de la variable dependiente 
esc my =medser(valacc) 
 ; calculo del coeficiente de determinación por dos procedimientos 
mat yy=y'y 
esc y2=escal(yy) 
mat yyestim=coef'x'y 
esc y2estim=escal(yyestim) 
esc rcuad2=1-(uu2 )/ (y2-nobs*pot(my,2)) 
esc rcuad21= (y2estim-nobs*pot(my,2))/ (y2-nobs*pot(my,2)) 
visesc rcuad2 
visesc rcuad21  
  
 ;CASO 304 
 ; creacion de variables en logaritmos 
ser lvalacc=ln(valacc) 
ser lvalcon=ln(valcon) 
  
 ;modelo lineal  
mc valacc 1 valcon 
esc aic=_aic 
 
 ;  modelo doblemente logaritmico 
mc lvalacc 1 lvalcon 
 ; calculo AIC 
esc aicl=_aic+2*_mvd 
 
 ; comparacion ajuste entre el modelo lineal y  el modelo doblemente logaritmico 
visesc aic 
visesc aicl 
  
 ; comparacion graficos de los dos modelos 
graf (valcon) l valacc 
graf (lvalcon) l lvalacc 
 
 
 
 ; CASOS 302/404 
 
deci 0 
 ;calculo de rendis y  representacion de series a precios corrientes 
ser rendis=cprn+anf 
graf rendis cprn  
 
 ; calculo de rendisd y representacion de series a precios constantes 
ser rendisd=rendis*cprnd/cprn 
graf rendisd cprnd  
 
 ; grafico de dispersion y estimacion MC 
graf (rendisd) cprnd 
mc cprnd 1 rendisd 
  
              ; calculo de coeficientes  
mat x ={ 1 rendisd} 
mat y= {cprnd} 
 
mat xx=x'x 
mat xy=x'y 
mat xxinv=xx$ 
mat coef =xxinv*xy 



mat coef1=(x'x)$x'y 
 
 ; calculo del vector de residuos 
mat u=y-x*coef1 
 
 ; calculo del estimador de la varianza de las perturbaciones 
mat uu=u'u 
esc uu2=escal(uu) 
mat uu1=y'y-coef'x'y 
esc uu22=escal(uu1) 
esc varu=uu2/(nobs-_nreg) 
esc desu=rcuad(varu) 
 
               ; calculo de la matriz var-cov de los estimadores 
mat covacoef=varu*(x'x)$ 
 
 ; calculo de la media de la variable dependiente 
esc my =medser(cprnd) 
 ; calculo del coeficiente de determinación por dos procedimientos 
mat yy=y'y 
esc y2=escal(yy) 
mat yyestim=coef'x'y 
esc y2estim=escal(yyestim) 
esc rcuad2=1-(uu2 )/ (y2-nobs*pot(my,2)) 
esc rcuad21= (y2estim-nobs*pot(my,2))/ (y2-nobs*pot(my,2)) 
visesc rcuad2 
visesc rcuad2 
 
  

; CASO 303 
 
 ; creacion de series 
ser lnlact=ln(lacteos) 
ser lnrdisp=ln(rdisp) 
ser invrdisp=1/rdisp 
 
 ; modelo 1 
mc lacteos 1 rdisp 
esc my =medser(lacteos) 
esc mx=medser(rdisp) 
esc promar1=_beta[2,1] 
visesc promar1 
esc elas1=_beta[2,1]*mx/my 
visesc elas1 
esc AIC1=_AIC 
visesc aic1 
 
 ; modelo 2 
mc lacteos 1 invrdisp 
esc promar2=-_beta[2,1]/(mx*mx) 
visesc promar2 
esc elas2=-_beta[2,1]/(mx*my) 
visesc elas2 
esc AIC2=_AIC 
visesc aic2 
 
 ; modelo 3 
mc lacteos 1 lnrdisp 
esc promar3=_beta[2,1]/(mx) 
visesc promar3 



esc elas3=_beta[2,1]/(my) 
visesc elas3 
esc AIC3=_AIC 
visesc aic3 
 
 ; modelo 4 
mc lnlact 1 lnrdisp 
esc promar4=_beta[2,1]*my/mx 
visesc promar4 
esc elas4=_beta[2,1] 
visesc elas4 
 ; calculo AICL 
esc aicl4=_aic+2*_mvd 
visesc aicl4 
 
 ; modelo 5 
mc lnlact 1 rdisp 
esc promar5=_beta[2,1]*my 
visesc promar5 
esc elas5=_beta[2,1]*mx 
visesc elas5 
 ; calculo AICL 
esc aicl5=_aic+2*_mvd 
visesc aicl5 
 
 ; modelo 6 
mc lnlact 1 invrdisp 
esc promar6=-_beta[2,1]*my/(mx*mx) 
visesc promar6 
esc elas6=-_beta[2,1]/mx 
visesc elas6 
 ; calculo AIC 
esc aicl6=_aic+2*_mvd 
visesc aicl6 
 
 ; calculo de los coeficientes 
mat x ={ 1 rdisp} 
mat y= {lacteos} 
mat xx=x'x 
mat xy=x'y 
mat xxinv=xx$ 
mat coef =xxinv*xy 
mat coef1=(x'x)$x'y 
 
 ; calculo del vector de residuos 
mat u=y-x*coef 
 
 ; calculo del estimador de la varianza de las perturbaciones 
mat uu=u'u 
esc u2=escal(uu) 
esc varu=u2/(nobs-_nreg) 
esc desu=rcuad(varu) 
visesc varu 
visesc desu 
              
  ; calculo de la matriz var-cov de los estimadores 
mat covacoef=varu*(x'x)$ 
 
 ; calculo de la media de la variable dependiente 
esc my =medser(lacteos) 



  
 ; calculo del coeficiente de determinación 
mat yy=y'y 
esc y2=escal(yy) 
esc rcuad2=1-(u2 )/ (y2-nobs*pot(my,2)) 
visesc rcuad2 
 
 
 ; CASO 305 
 
 ; estimacion modelo MC 
mc ventas 1 publ incent 
 
 ; calculo de los coeficientes 
mat x ={ 1 publ incent} 
mat y= {ventas} 
mat xx=x'x 
mat xy=x'y 
mat xxinv=xx$ 
mat coef =xxinv*xy 
mat coef1=(x'x)$x'y 
 
 ; calculo del vector de residuos 
mat u=y-x*coef 
 
 ; calculo del estimador de la varianza de las perturbaciones 
mat uu=u'u 
esc u2=escal(uu) 
esc varu=u2/(nobs-3) 
esc desu=rcuad(varu) 
               
  ; calculo de la matriz var-cov de los estimadores 
mat covacoef=varu*(x'x)$ 
 
 ; calculo de la media de la variable dependiente 
esc my =medser(ventas) 
 
 ; calculo del coeficiente de determinación 
mat yy=y'y 
esc y2=escal(yy) 
esc rcuad2=1-(u2 )/ (y2-nobs*pot(my,2)) 
 
 ; creacion variables en logaritmos 
ser lventas=ln(ventas) 
ser lpubl=ln(publ) 
ser lincent=ln(incent) 
 
 ; modelo 1 
mc ventas 1 publ  
esc aic1=_aic 
 
 ; modelo 2 
mc ventas 1 incent 
esc aic2=_aic 
 
 ; modelo 3 
mc ventas 1 publ incent 
esc aic3=_aic 
 
 ; modelo 4 



mc ventas 1 lpubl incent 
esc aic4=_aic 
 
 ; modelo 5 
mc ventas 1 publ lincent 
esc aic5=_aic 
 
 ; modelo 6 
mc ventas 1 lpubl lincent 
esc aic6=_aic 
 
 ; modelo 7 
mc lventas 1 publ incent 
 ; calculo AIC 
esc aicl7=_aic+2*_mvd 
 
 
 ; modelo 8 
mc lventas 1 lpubl lincent 
 ; calculo AIC 
esc aicl8=_aic+2*_mvd 
 
 
 ; visualizacion coeficientes AIC 
visesc aic1 
visesc aic2 
visesc aic3 
visesc aic4 
visesc aic5 
visesc aic6 
visesc aicl7 
visesc aicl8 
 
  

; CASO 306/411 
   
 ; creacion de variables  en logaritmos 
ser lvabpri=ln(vabpri) 
ser lkpri=ln(kpri) 
ser lempri=ln(empri) 
 
 
 ; estimacion modelo 
mc lvabpri 1 lkpri lempri 
 
 
 ; contraste rendimientos constantes a escala 
  ; Ho: beta(2)+beta(3)=1 
CHN ir  1 
CHN mr 1  beta(2)+beta(3)=1 
CHN fr 1 
  
 
 ; CASO 307 
 
 ; la trampa de las variables ficticias 
ser hombre=sexo 
ser mujer = 1-sexo 
mc absen 1 hombre mujer 
  



 ; soluciones a la trampa de las variables ficiticias 
  ; prescindir del termino independiente 
mc absen  hombre mujer 
  ; prescindir de una categoria en la v. ficticia 
mc absen  1 hombre 
  
 ; modelo en funcion salario y sexo(hombre) 
mc absen 1 salario sexo 
ser absen1 = _beta[1,1] +  _beta[2,1]*salario + _beta[3,1]*sexo 
graf(salario) absen absen1 
 
 ; modelo en funcion salario y sexo(hombre) con interaccion 
ser sexsal=sexo*salario 
mc absen 1 salario sexo sexsal 
 
 ; modelo completo 
mc absen 1 taller sexo edad antigue salario 
 
 
 ; CASOS 309/407 (Datos CASO 302) 
 ; estimacion funcion de consumo de Broawn 
 
 ;calculo de rendis y rendisd 
ser rendis=cprn+anf 
ser rendisd=rendis*cprnd/cprn 
  
 
 ; modelo de Brown 
mc cprnd 1 rendisd cprnd(-1)  
  
 ; contraste H nula: pmlp=0.90 
  ; Ho: beta(2)+0.9*beta(3)=0.9 
CHN ir  1 
CHN mr 1 beta(2)+0.9*beta(3)=0.9 
CHN fr 1 
             
 ; contraste de estabilidad estructural 
   ; F=0 en (1954-1975) y F=1 en (1976-1991) 
      ; a. utilizando variables ficticias 
  ;  creacion de nuevas variables 
muest 1955 1991 
ser rendisdf=rendisd*f1 
ser cprnd1f=cprnd(-1)*F1 
 
  ; estimacion del modelo con variables ficticias 
 
mc cprnd 1 rendisd cprnd(-1) F1 rendisdf  cprnd1f 
  ; contraste 
CHN ir  3  
CHN mr 1 beta(4)=0 
CHN mr 2 beta(5)=0 
CHN mr 3 beta(6)=0 
CHN fr 3 
 
 
 ; b. utilizando suma de cuadrados de residuos 
  ; estimacion modelo general 
muest 1955 1991 
mc cprnd 1 rendisd cprnd(-1) F1 rendisdf  cprnd1f 
esc scrg=_scr 



 
  ; estimacion modelo restringido 
muest 1955 1991 
mc cprnd 1 rendisd cprnd(-1)  
 
  ; contraste 
esc scrr=_scr 
esc estF=((scrr-scrg)/3)/(scrg/(36-6)) 
visesc estF 
esc nscF= probF(estF,3,30) 
visesc nscF 
 

; ?existen efectos de interaccion entre la variable ficticia y las variables explicativas? 
  ; a. utilizando variables ficticias 
  ; estimacion del modelo con variables ficticias 
 
mc cprnd 1 rendisd cprnd(-1) F1 rendisdf  cprnd1f 
  ; contraste 
CHN ir  2 
CHN mr 1 beta(5)=0 
CHN mr 2 beta(6)=0 
CHN fr 2 
 
 ; b. utilizando suma de cuadrados de residuos 
  ; estimacion modelo general 
muest 1955 1991 
mc cprnd 1 rendisd cprnd(-1) F1 rendisdf  cprnd1f 
esc scrg=_scr 
 
  ; estimacion modelo restringido 
muest 1955 1991 
mc cprnd 1 rendisd cprnd(-1) F1 
 
  ; contraste 
esc scrr=_scr 
esc estF=((scrr-scrg)/2)/(scrg/(36-6)) 
visesc estF 
esc nscF= probF(estF,2,30) 
visesc nscF 
 
 
 ; CASO 310/412 
 
muest 1907 1960 
mc ventas 1 gpub ventas(-1) F1 F2 F3 
 
 
CHN ir  1 
CHN mr 1 1.06*beta(2)+1.0875*beta(3)=1.15275 
CHN fr 1 
esc estf=_forlarh 
esc nscf=_estfh 
visesc estf 
visesc nscf 
 
  

; CASO 311 
 ; simulacion de Montecarlo de la funcion de consumo 
 
 ; simulacion de una serie 



ser cons=2+0.85*rendis+1.2*rndn 
mc cons 1 rendis 
 
 ; simulacion de 10 series  
 
  ; inicializacion 
esc alfa =0 
esc beta=0 
 
; de 1 a 10 series 
repite  j desde 1 hasta 10 
ser cons=2+0.85*rendis+1.2*rndn 
mc cons 1 rendis 
esc  alfa=alfa+_beta[1,1] 
esc  beta= beta+_beta[2,1] 
finrepite 
 
 ; estimaciones medias 
esc alfam=alfa/10 
esc betam=beta/10 
visesc alfam 
visesc betam 
 
 ; sesgos 
esc sesgoalf=2-alfam 
esc sesgobet=0.85-betam 
visesc sesgoalf 
visesc sesgobet 
 
 
 ; CASO 403a 
 
ser rbmad92=(ibmad92-ibmad92(-1))/ibmad92(-1) 
mc rbmad92 1 rbmad92(-1) 
 
  
 ; CASO 403b 
 
ser rbtok92=(ibtok92-ibtok92(-1))/ibtok92(-1) 
mc rbtok92 1 rbtok92(-1) 
 
 
 ; CASO 406 
 
ser rbtok92=(ibtok92-ibtok92(-1))/ibtok92(-1) 
ser rbmad92=(ibmad92-ibmad92(-1))/ibmad92(-1) 
mc rbmad92 1 rbtok92 
 
 
 ; CASO 408 
 
 ; calculo indices de precios 
ser ppesca=pesca*100/pescad 
ser pcarne =carne*100/carned 
graf ppesca pcarne 
  
 
 ; creacion de variables  en logaritmos 
ser lpescad=ln(pescad) 
ser lppesca=ln(ppesca) 



ser lpcarne=ln(pcarne) 
ser lcprnd=ln(cprnd) 
 
 ; estimacion modelo 
mc lpescad 1 lppesca lpcarne lcprnd 
 
 ; contraste restriccion homogeneidad 
  ; Ho: beta(2)+beta(3)+beta(4)=0 
CHN ir  1 
CHN mr 1  beta(2)+beta(3)+beta(4)=0 
CHN fr 1 
            
 
 ; CASO 701 (Datos en CASO 301) 
 
 ; modelo lineal 
muest 1 20 
deci 0 
  ; grafico de dispersion y estimacion 
graf (valcon)  valacc 
mc valacc 1 valcon 
 
  ; contraste White 
CWH   PC RA   
   
  ; contraste de Golfeld y Quandt 
CGQ (valcon) 1 7 14 20 
 
  ; matriz de covarianzas consistente bajo heteroscedasticidad 
muest 1 20 
mc (ch) valacc 1 valcon  
 
     
 ;  modelo doblemente logaritmico 
  ; creacion de variables en logaritmos 
muest 1 20 
ser lvalacc=ln(valacc) 
ser lvalcon=ln(valcon) 
  
  ; grafico de dispersion y estimacion 
graf (lvalcon) l lvalacc 
mc lvalacc 1 lvalcon 
  
  ; contraste White 
CWH   PC RA   
 
  ; contraste de Golfeld y Quandt 
CGQ (valcon) 1 7 14 20 
 
  ; matriz de covarianzas consistente bajo heteroscedasticidad 
muest 1 20 
mc (ch) lvalacc 1 lvalcon  
  
  
  
 

; CASO 702 
 
 ;  modelo doblemente logaritmico 
  ; creacion de variables en logaritmos 



ser lnhostel=ln(hostel) 
ser lnrdisp=ln(rdisp) 
 
  ; estimacion MC 
muest 1 40 
mc lnhostel 1 lnrdisp estmed estsup tamhog 
 
  ; contraste White 
CWH   RA   
 
  ; contraste de Golfeld y Quandt 
CGQ (rdisp) 1 15 26 40 
 
  ; matriz de covarianzas consistente bajo heteroscedasticidad 
muest 1 40 
mc (ch)  lnhostel 1 lnrdisp estmed estsup tamhog;  modelo doblemente logaritmico 
  


